# $Id$
# Maintainer:
# Contributor: Jan de Groot <jgc@archlinux.org>

pkgname=gdome2
pkgver=0.8.1
pkgrel=5
pkgdesc="Gnome DOM Engine"
arch=('i686' 'x86_64')
url="http://gdome2.cs.unibo.it/"
license=('LGPL')
depends=('libxml2' 'glib2')
source=("http://gdome2.cs.unibo.it/tarball/${pkgname}-${pkgver}.tar.gz"
        gdome2-0.8.1-libxml2.patch)
md5sums=('bfc114e59eec50cbda8e4ece751ff022'
         'bde155107f30bf65d441adf70efc3c00')

prepare() {
  cd ${pkgname}-${pkgver}
  patch -p0 -i "${srcdir}/gdome2-0.8.1-libxml2.patch"
}


build() {
  cd ${pkgname}-${pkgver}
  ./configure --prefix=/usr --mandir=/usr/share/man
  make
}

package() {
  cd ${pkgname}-${pkgver}
  make DESTDIR="${pkgdir}" install
}
